

MySQL Protocol Tutorial

Stéphane Legrand <stephleg@free.fr>

November 21, 2020

Abstract

This tutorial illustrates the use of the MySQL Protocol library, a native OCaml implementation of the MySQL/MariaDB client protocol.

You can install this library with OPAM. The source code is available in the MySQL Protocol home page.

Contents

1	Introduction	4
2	Modules	4
3	Helper functions	4
4	Configuration	5
5	Connection	6
6	Select database	6
7	Non prepared statement	6
7.1	Create	6
7.2	Execute	6
7.3	Get result	7
7.3.1	Result without record	7
7.3.2	Result with records	7
8	Prepared statement	7
8.1	Create	7
8.2	Prepare	8
8.3	Execute	8
8.3.1	Simple execute	8
8.3.2	With parameters	8
8.3.3	With cursor	9
8.4	Get result	9
8.4.1	Result without record	9
8.4.2	Result with records	9
8.5	Close statement	10
9	Procedure call	10
10	Ping server	12
11	Change user	12
12	Reset connection	12
13	Reset the session	12
14	Catching errors	12

15 Disconnect

13

1 Introduction

The library has to be installed before using the code below. Please use the OPAM package manager. The OCaml source code of this tutorial is available in the *examples* directory.

2 Modules

First, some convenient alias for the modules used.

```
module Mp_client = Mysql_protocol.Mp_client
module Mp_data = Mysql_protocol.Mp_data
module Mp_execute = Mysql_protocol.Mp_execute
module Mp_result_set_packet = Mysql_protocol.Mp_result_set_packet
module Mp_capabilities = Mysql_protocol.Mp_capabilities
```

3 Helper functions

We define one function to print the result of SQL statements like INSERT, UPDATE, GRANT...

```
let print_result sql r =
  print_endline ("Result of the SQL statement \""
    ^ sql ^ "\": \n "
    ^ (Mp_client.dml_dcl_result_to_string r) ^ "\n")
```

And two others to print the result of SELECT SQL statements.

```
let print_row fields row =
  let print_data f =
    let (field_name, field_pos) = f in
    let data = List.nth row field_pos in
    print_endline (" " ^ field_name ^ ": "
      ^ (Mp_data.to_string data))
  in
  let () = List.iter print_data fields in
  print_endline " -- -- "

let print_set sql r =
  let (fields, rows) = r.Mp_result_set_packet.rows in
  let () = print_endline (
```

```

    "Result set for the SQL statement \"" ^ sql ^ "\": \n") in
  let print_rows =
    let () = List.iter (print_row fields) rows in
    print_newline ()
  in
  print_rows

```

4 Configuration

To be able to connect to the database server, we first have to configure the client.

```

let addr = Unix.inet_addr_loopback
let port = 3306
let sockaddr = Unix.ADDR_INET(addr, port)
(* let sockaddr = Unix.ADDR_UNIX "/path/to/file.sock" in *)

let db_user = "user_ocaml_ocmp"
let db_password = "ocmp"
let db_name = "test_ocaml_ocmp_utf8"

```

The database server is listening on the loopback interface and it uses the standard port. The login to authenticate to the server is "user_ocaml_ocmp" with the password "ocmp". And we will use the "test_ocaml_ocmp_utf8" database.

Now, we can create a configuration.

```

let config = Mp_client.configuration ~user:db_user
~password:db_password ~sockaddr:sockaddr
~databasename:db_name () in

```

Here, the default charset and collation is used. The default value for charset is Utf8 and the default value for collation is Utf8_general_ci. If you want to specify an other value, you can use the charset parameter. For instance:

```

let config = Mp_client.configuration ~user:db_user
~password:db_password ~sockaddr:sockaddr
~databasename:db_name
~charset:(Mp_charset.Latin1, Mp_charset.Latin1_swedish_ci)
() in

```

To have the complete list of available charset and collation, you can read the documentation of the Mp_Charset module.

5 Connection

Now, we can connect our client to the database server.

```
| let connection = Mp_client.connect  
| ~configuration:config () in
```

By default, the connection is not initialized right after the call to the `connect()` function. It's delayed until necessary (ie until the first real request). You can immediately force the connection by using the `force` parameter:

```
| let connection = Mp_client.connect  
| ~configuration:config ~force:true () in
```

6 Select database

To specify the current database, use the following function:

```
| let () = Mp_client.use_database  
| ~connection:connection ~databasename:db_name in
```

7 Non prepared statement

A non prepared statement is the simplest way to send a statement to the server. If your statement doesn't have any parameter (ie is a static string) and is used only a few times, it's usually sufficient.

WARNING: You SHOULD NOT use a non prepared statement if it contains a parameter with non trusted value.

7.1 Create

The first step is to create the statement from the SQL string.

```
| let sql =  
| "INSERT INTO ocmp_table (col1, col2) VALUES ('col1', 123.45)"  
in  
| let stmt = Mp_client.create_statement_from_string sql in
```

7.2 Execute

Next, we send the statement to the server to execute it.

```
| let r = Mp_client.execute ~connection:connection  
| ~statement:stmt () in
```

7.3 Get result

After being executed, the statement result can be retrieved.

7.3.1 Result without record

For statement which returns only a simple result without any record (INSERT, UPDATE, DELETE, GRANT... statement), you can use the `get_result_ok()` function.

```
| let r = Mp_client.get_result r in  
| let r = Mp_client.get_result_ok r in
```

To print this result, use the `print_result()` helper function.

```
| let () = print_result sql r in
```

7.3.2 Result with records

For statement which returns records (typically SELECT statement), you can use the `get_result_set()` function.

```
| let r = Mp_client.get_result r in  
| let r = Mp_client.get_result_set r in
```

To print this result, use the `print_set()` helper function.

```
| let () = print_set sql r in
```

8 Prepared statement

Especially when the statement includes some parameters, you should use a prepared statement. The parameters values will then be correctly enclosed in the statement by the database server and all special characters will be automatically escaped. You can of course also use a prepared statement even if the statement doesn't have any parameter.

8.1 Create

The first step is the same as for a non prepared statement, you have to create the statement from the SQL string with the same function.

```
| let sql = "SELECT * FROM ocmp_table WHERE col1=?" in  
| let stmt = Mp_client.create_statement_from_string sql in
```

8.2 Prepare

Then, you prepare the statement.

```
| let prep = Mp_client.prepare ~connection:connection  
| ~statement:stmt in
```

Once a statement has been prepared, you can execute it several times without calling the `prepare()` function again.

8.3 Execute

You execute a prepared statement with the same function as for a non prepared one. Nonetheless, for a prepared statement, the `execute()` function may accept more parameters.

8.3.1 Simple execute

For the simplest use case (no parameter in the statement, no cursor), you execute the statement as for a non prepared one.

```
| let r = Mp_client.execute ~connection:connection  
| ~statement:prep () in
```

8.3.2 With parameters

If you have some parameters in the statement, you first need to create the list of these parameters in the same order of appearance as in the statement. Please see the documentation for the `Mp_data` module to have a complete list of data constructor and learn which one to use for each column types.

```
| let params = [  
|   Mp_data.data_varstring "col2";  
|   Mp_data.data_decimal (Num.num_of_string "98765/100")  
| ] in
```

And you add the `params` function parameter for the execution.

```
| let r = Mp_client.execute ~connection:connection  
| ~statement:prep ~params:params () in
```


8.3.3 With cursor

By default, no cursor is used when a prepared statement is executed. So the server will always return all the corresponding records. If you want to be able to fetch the result by parts (record by record for instance), you need to specify the cursor option in the `execute()` function.

```
| let stmt = Mp_client.execute ~connection:connection  
| ~statement:prep ~params:params  
| ~flag:Mp_execute.Cursor_type_read_only () in
```

WARNING: For now, only `Cursor_type_read_only` type is supported.

8.4 Get result

After being executed, the statement result can be retrieved.

8.4.1 Result without record

For statement which returns only a simple result without any record (INSERT, UPDATE, DELETE, GRANT... statement), there is no difference compared with the non prepared statements. You can also use the `get_result_ok()` function.

```
| let r = Mp_client.get_result r in  
| let r = Mp_client.get_result_ok r in
```

To print this result, use the `print_result()` helper function.

```
| let () = print_result sql r in
```

8.4.2 Result with records

For statement which returns records (typically SELECT statement), if you haven't used a cursor, you cannot use `fetch`. So you will retrieve all the rows with the same method as a non prepared statement.

```
| let r = Mp_client.get_result r in  
| let r = Mp_client.get_result_set r in
```

To print this result, use the `print_set()` helper function.

```
| let () = print_set sql r in
```

If you have used a cursor, you have to use fetch to retrieve the records. By default, the fetch() function get one record at each call. To specify an other number, use the nb_rows function parameter.

```
let stmt = Mp_client.execute ~connection:connection
~statement:prep ~params:params
~flag:Mp_execute.Cursor_type_read_only () in
let () =
  try
    while true do
      let rows = Mp_client.fetch
        ~connection:connection
        ~statement:stmt ()
      in
      let rows = Mp_client.get_fetch_result_set
        rows
      in
      print_set sql rows
    done
  with
  | Mp_client.Fetch_no_more_rows -> ()
in
```

8.5 Close statement

When a prepared statement has become useless (ie you don't need to execute it again), you can and should destroy it.

```
let () = Mp_client.close_statement
~connection:connection ~statement:prep in
```

9 Procedure call

First we create the stored procedure after checking if it was already defined.

```
let sql = "DROP PROCEDURE IF EXISTS ocmp_proc" in
let stmt = Mp_client.create_statement_from_string sql in
let _ = Mp_client.execute ~connection:connection
~statement:stmt ()
in
let sql =
```

```

    "CREATE PROCEDURE ocmp_proc() "
    ~ "BEGIN SELECT * FROM ocmp_table; END"
in
let stmt = Mp_client.create_statement_from_string
    sql in
let r = Mp_client.execute ~connection:connection
    ~statement:stmt ()
in
let r = Mp_client.get_result r in
let r = Mp_client.get_result_ok r in
let () = print_result sql r in

```

Next we call the stored procedure. The result type of `execute()` is `list` result so we use the helper function `f()` to extract the value we need.

WARNING: To use stored procedures, the capabilities defined for the connection configuration **MUST** include the capability named `Client_multi_results`. This is the default.

```

let sql = "CALL ocmp_proc()" in
let stmt = Mp_client.create_statement_from_string
    sql
in
let r = Mp_client.execute ~connection:connection
    ~statement:stmt ()
in
let r = Mp_client.get_result_multiple r in
let f e =
    try
        let rs = Mp_client.get_result_set e in
        print_set sql rs
    with
    | Failure _ ->
        let rs = Mp_client.get_result_ok e in
        let affected_rows = rs.Mp_client.affected_rows in
        print_endline (Printf.sprintf
            "Result OK: affected rows=%Ld" affected_rows)
in
let () = List.iter f r in

```

10 Ping server

To avoid a timeout or to test the connection, you can send a ping to the server. No result is returned but an `Mp_client.Error` exception can be raised.

```
| let () = Mp_client.ping ~connection:connection in
```

11 Change user

Connect to the same server with a different user and/or database.

```
| let _ = Mp_client.change_user  
| ~connection:connection  
| ~user:"db_user"  
| ~password:"db_password"  
| ~databasename:"db_name" ()  
in
```

12 Reset connection

Reset connection without re-authentication. This is useful if you need to destroy the session context (temporary tables, session variables, etc.) in the database server.

```
| let () = Mp_client.reset_connection  
| ~connection:connection in
```

13 Reset the session

Disconnect and reconnect.

```
| let () = Mp_client.reset_session  
| ~connection:connection in
```

14 Catching errors

Whenever the database server returns an error, an `Mp_client.Error` exception is raised.

```
| let stmt = Mp_client.create_statement_from_string  
| ("BAD SQL QUERY") in
```

```

let () =
  try
    let _ = Mp_client.execute
      ~connection:connection ~statement:stmt () in
    ()
  with
  | Mp_client.Error error ->
    print_newline ();
    print_endline ("Catch error, exception is: "
      ^ (Mp_client.error_exception_to_string error))
in

```

15 Disconnect

To close the connection to the server, use the `disconnect()` function.

```

let () = Mp_client.disconnect ~connection:connection in

```